p GetPhonesByMem  B                   Export                                     

 ******************************************************************************* 

 /Define cpy_getPhonesByMem_Interface                                            

 /copy Qrpglesrc,CB.Phones                                                       

 /Undefine cpy_getPhonesByMem_Interface                                          

 /free                                                                           

         Clear OutputDs2;                                                        

         Clear d@PhoneOut;                                                       

         TrueFalse='true';                                                       

         SetSOAPPhByMem(%char(MemberNum));                                       

         procedure='GetPhonesByMem';                                             

         dou retcod='C' or retcod='c';                                           

           active=getActiveFlg(procedure);                                       

           if active='N';                                                        

             retcod=sndrcverr('SRV0004':procedure);                              

             wrtWSLogs(procedure:p@jobn:p@user:p@job#:p@prog:MemberNum:0:
                                       'SRV0004':'   ':SOAP:SOAP2);                            

            OutputDs.errorId='SRV0004';                                                        

            OutputDs.errorMsg=rtvmsgD('SRV0004':procedure);                                    

            pPhonesByMem=%addr(OutputDs);                                                      

            return  pPhonesByMem;                                                              

          else;                                                                                

            leave;                                                                             

          endif;                                                                               

        enddo;                                                                                 

 // The http_setCCSIDs() routine tells HTTPAPI to translate my XML document from EBCDIC/UTF-8  

 // unicode, because this is what the web service is expecting. 1208 is the CCSID for UTF-8    

 // unicode. 0 is a special value that means "use my jobs default CCSID," which will be EBCDIC 

 // in my environment.                                                                         

 http_setCCSIDs(1208: 0);                                                                      

 // The http_debug() routine tells HTTPAPI to produce a log containing debugging information,  

 // which can be invaluable in troubleshooting HTTPAPI.                                        
  // The debug file will be placed in the IFS under the name /tmp/httpapi_debug.txt.     

        http_debug(*ON);                                                                 

  SoapAct='SOAPACT';                                                                     

  http_xproc( HTTP_POINT_ADDL_HEADER                                                     

            : %paddr(Add_SOAPACTION) );                                                  

        open(e) mbfflcmtdt;                                                              

        Ep =   'http://' +                                                               

               %trim(GetValue('WEBSERVICE':'ENDPOINT'))+                                 

               Quote;                                                                    

  dou retcod='C' or retcod='c';                                                          

        error=*off;                                                                      

        rc = http_post_xml( Ep                                                           

                          : %addr(SOAP) + VARYINGDATAOFFSET                              

                          : %len(SOAP)                                                   

                          : *NULL                                                        
                         : %paddr(MapXmlData)                       

                         : %addr(xml_value)                         

                         : HTTP_TIMEOUT                             

                         : HTTP_USERAGENT                           

                         : 'text/xml; charset=UTF-8');              

 select;                                                            

  when (rc=1);                                                      

  if  error=*on and OutputDs2(error_elem).ErrorId='ERR0476';        

     retcod=sndrcverr('ERR0476':procedure);                         

     wrtWSLogs(procedure:p@jobn:p@user:p@job#:p@prog:MemberNum:0:   

                   OutputDs2(error_elem).ErrorId:                   

                   OutputDs2(error_elem).ErrorMsg:SOAP:SOAP2);      

  else;                                                             

    leave;                                                          

  endif;                                                            

  when rc=0;                                                        

      retcod=sndrcverr('SRV0001':procedure);                        

      SOAP2=*blanks;                                                
     wrtWSLogs(procedure:p@jobn:p@user:p@job#:p@prog:MemberNum:0:  

                             'SRV0001':'   ':SOAP:SOAP2);          

     OutputDs2(1).errorId='SRV0001';                               

     OutputDs2(1).errorMsg=rtvmsgD('SRV0001':'  ');                

 when rc=-1;                                                       

     retcod=sndrcverr('SRV0003':procedure);                        

     SOAP2=*blanks;                                                

     wrtWSLogs(procedure:p@jobn:p@user:p@job#:p@prog:MemberNum:0:  

                             'SRV0003':'   ':SOAP:SOAP2);          

     OutputDs2(1).errorId='SRV0003';                               

     OutputDs2(1).errorMsg=rtvmsgD('SRV0003':'  ');                

 when rc>1;                                                        

     retcod=sndrcverr('SRV0002':procedure);                        

     SOAP2=*blanks;                                                

     wrtWSLogs(procedure:p@jobn:p@user:p@job#:p@prog:MemberNum:0:  

                             'SRV0002':'   ':SOAP:SOAP2);          

     OutputDs2(1).errorId='SRV0002';                               

     OutputDs2(1).errorMsg=rtvmsgD('SRV0002':'  ');                

 endsl;                                                            
 enddo;                                                                  

         if primary_elem=0;                                              

         primary_elem=1;                                                 

         endif;                                                          

         OutputDs.contact=OutputDs2(primary_elem).contact;               

         OutputDs.CPhones=OutputDs2(primary_elem).Cphones;               

         OutputDs.CPhones2=OutputDs2(primary_elem).Cphones2;             

         OutputDs.errorId=OutputDs2(primary_elem).errorId;               

         OutputDs.errorMsg=OutputDs2(primary_elem).errorMsg;             

         wrtWSLogs(procedure:p@jobn:p@user:p@job#:p@prog:MemberNum:0:    

                       OutputDs.ErrorId:OutputDs.ErrorMsg:SOAP:SOAP2);   

         pPhonesByMem=%addr(OutputDs);                                   

         return  pPhonesByMem;                                           

         *inlr = *on;                                                    
  /end-free                                  

 P                 e                                 
